* SPSSS-SYNTAX zur Zuweisung von Fällen zu den Clustern/Bindungsmustern

* 1. Berechnen der BFPE-Skalenwerte

* 1a. Umkodieren der umgekehrt gepolten Items

RECODE

 bfpe04 bfpe05 bfpe06 bfpe10 bfpe12 bfpe23 bfpe25  (0=4)  (1=3)  (3=1)  (4=0)  .

EXECUTE .

* 1b. Berechnen der Skalenwerte (mittlere Itemwerte)

COMPUTE akzprob = (bfpe03 + bfpe09 + bfpe13 + bfpe17 + bfpe20 + bfpe21 +

  bfpe24 + bfpe27 + bfpe28 + bfpe29 + bfpe31)/11 .

EXECUTE .

COMPUTE oeffber = (bfpe02 + bfpe04 + bfpe05 + bfpe06 + bfpe07 + bfpe10 +

  bfpe12 + bfpe15 + bfpe23 + bfpe25 + bfpe30)/11 .

EXECUTE .

COMPUTE zuwbed = (bfpe08 + bfpe11 + bfpe14 + bfpe16 + bfpe18 + bfpe19 +

  bfpe22 + bfpe26)/8 .

EXECUTE .

* 1c. Wiederherstellung der ursprünglichen Daten

RECODE

 bfpe04 bfpe05 bfpe06 bfpe10 bfpe12 bfpe23 bfpe25  (0=4)  (1=3)  (3=1)  (4=0)  .

EXECUTE .

* 1d. Berechnen der z-Werte

COMPUTE zakzprob = (akzprob-1.17)/0.69 .

EXECUTE .

COMPUTE zoeffber = (oeffber-2.67)/0.66 .

EXECUTE .

COMPUTE zzuwbed = (zuwbed-2.18)/0.72 .

EXECUTE .

* 2. Berechnung der Diskriminanzfunktionswerte

COMPUTE f1 = (-1.354*zakzprob + 1.189*zoeffber + 0.377*zzuwbed)+0.010 .

EXECUTE .

COMPUTE f2 = (0.281*zakzprob + 0.024*zoeffber + 1.549*zzuwbed)+0.020 .

EXECUTE .

COMPUTE f3 = (1.259*zakzprob + 1.306*zoeffber - 0.464*zzuwbed) - 0.009 .

EXECUTE .

* 3. Bestimmung der Distanzwerte Dis zu den Zentroiden der fünf Cluster

COMPUTE DCL1 = SQRT ((-1.447-f1)**2+(-1.389-f2)**2+(-0.332-f3)**2) .

EXECUTE .

COMPUTE DCL2 = SQRT ((1.859-f1)**2+(-1.567-f2)**2+(0.263-f3)**2) .

EXECUTE .

COMPUTE DCL3 = SQRT ((2.550-f1)**2+(1.336-f2)**2+(-0.052-f3)**2 ).

EXECUTE .

COMPUTE DCL4 = SQRT ((-0.174-f1)**2 +(1.079-f2)**2+(-0.150-f3)**2) .

EXECUTE .

COMPUTE DCL5 = SQRT ((-3.318-f1)**2+(0.749-f2)**2+(0.327 - f3)**2) .

EXECUTE .

* 4. Zuweisung der Fälle zum nächstgelegenen Clusterzentroid

IF ((DCL1)<(DCL2) and (DCL1)<(DCL3) and (DCL1)<(DCL4) and (DCL1)<(DCL5)) CluNr = 1 .

EXECUTE .

IF ((DCL2)<(DCL1) and (DCL2)<(DCL3) and (DCL2)<(DCL4) and (DCL2)<(DCL5)) CluNr = 2 .

EXECUTE .

IF ((DCL3)<(DCL1) and (DCL3)<(DCL2) and (DCL3)<(DCL4) and (DCL3)<(DCL5)) CluNr = 3 .

EXECUTE .

IF ((DCL4)<(DCL1) and (DCL4)<(DCL2) and (DCL4)<(DCL3) and (DCL4)<(DCL5)) CluNr = 4 .

EXECUTE .

IF ((DCL5)<(DCL1) and (DCL5)<(DCL2) and (DCL5)<(DCL3) and (DCL5)<(DCL4)) CluNr = 5 .

EXECUTE .

